

# データ解析（国際開発）

## (6) スペクトログラム（実習）

高田潤一

takada@ide.titech.ac.jp

2010 年 4 月 30 日

### 概要

本日は、残っている短時間フーリエ変換の説明の後に、スペクトログラムの実習を行う。

## 1 試料

解析に用いる対象は音声信号とする。Scilab では wav ファイルをデータとして入出力できる。そこで、フリーのナレーション音源（mp3 形式）を [1] よりダウンロードし、SCMPX [2] で wav ファイルに変換したものを OCW-i にアップロードしている。自分で各種の音源をダウンロードして試しても構わないが、長いデータはメモリサイズの関係で扱うことができないため、数秒程度を切り出して処理することを勧める。

### 1.1 チャープ信号

短時間フーリエ変換のような時間周波数解析を議論する際によく用いられる信号としてチャープ信号がある。チャープ信号は、周波数が時間の関数として変化する信号であり、自己相関関数がデルタ関数状となるため、レーダのように伝搬遅延時間を測定する用途に用いられる。数式で表すと、

$$x(t) = \sin 2\pi(f_{\min} + \Delta f t)t \quad (1)$$

のように表される。なお、瞬時周波数  $f(t)$  は位相成分の微分を用いて、

$$f(t) = \frac{d \arg x(t)}{2\pi dt} = f_{\min} + \frac{1}{2}\Delta f t \quad (2)$$

で表される。

プログラム chirp.sce では、チャープ信号のパラメタを設定して wav ファイルを出力する。

```
// Chirp signal for wave file
clear;
tmax = 0.1; // time in second
fs = 44100; // sampling rate in Hz
nbits = 16; // quantization bits
t = soundsec(tmax,fs); // time sample in second
```

```

fmin = 100; // minimum frequency of chirp signal in Hz
fmax = 3000; // maximum frequency of chirp signal in Hz
df = 0.5 * (fmax - fmin) / tmax; // gradient of frequency in Hz/second
x = sin(2*pi*t.*(fmin+t*df)); // chirp signal
wavwrite(x, fs, nbits, 'chirp.wav'); // write to wav file

```

なお、同様に正弦波のパラメタを設定して wav ファイルを出力する sine.sce も準備した。

```

// sine signal for wave file
clear;
tmax = 0.1; // time in second
fs = 44100; // sampling rate in Hz
nbits = 16; // quantization bits
t = soundsec(tmax,fs); // time sample in second
f0 = 1000; // frequency of sine signal in Hz
x = sin(2*pi*t.*f0); // chirp signal
wavwrite(x, fs, nbits, 'sine.wav'); // write to wav file

```

各関数、コマンドの詳細については、ヘルプファイルを参照のこと。

## 2 スペクトログラム

Scilab にはスペクトログラムを表示する組み込み命令として mapsound コマンドが実装されている。ソースコードは Scilab ディレクトリの下にある macros/sound/mapsound.sci である。

```

// Scilab ( http://www.scilab.org/ ) - This file is part of Scilab
// Copyright (C) 1990 - INRIA - Scilab
//
// This file must be used under the terms of the CeCILL.
// This source file is licensed as described in the file COPYING, which
// you should have received as part of this distribution. The terms
// are also available at
// http://www.cecill.info/licences/Licence_CeCILL_V2-en.txt

function mapsound (w,dt,fmin,fmax,simpl,rate)
// Plots a sound map for a sound.
// It does FFT at time increments dt.
// rate is the sampling rate.
// simpl points are collected for speed reasons.

[lhs,rhs]=argn(0);
if ( rhs <= 5 ) then rate=0;end
if ( rhs <= 4 ) then simpl=1; end ;
if ( rhs <= 3 ) then fmax=1500; end ;

```

```

if ( rhs <= 2 ) then fmin=100; end ;
if ( rhs <= 1 ) then dt=0.1; end ;
n=prod(size(w));
defaultrate=22050;
if rate==0; rate=defaultrate; end;
dp=dt*rate;
points=2^floor(log(dp)/log(2));
ind=fmin/rate*points/simpl:fmax/rate*points/simpl;
f=abs(fft(w(1:points),1));
f=sum(matrix(f,points/simpl,simpl),'c')';
M=f(ind);
i=1;
i=i+dp;
while i+points <= n;
    f=abs(fft(w(i:i+points-1),1));
    f=sum(matrix(f,points/simpl,simpl),'c')';
    M=[M;f(ind)];
    i=i+dp;
end;
[nl,nc]=size(M);
if nl==1 ; return;end
x_abs= (0:(nl-1))*(n/rate)/(nl-1);
y_abs= (0:(nc-1))/(nc-1);
y_abs= fmin*(1-y_abs) + fmax*(y_abs)
grayplot(x_abs,y_abs,-M)
endfunction

```

これを用いて , wav ファイルからスペクトログラムを生成するプログラム spectrogram.sci が以下のように書ける .

```

clear // reset variables
clf() // reset graphs
filename = "chirp.wav"; // wav file name
// filename = "sine.wav"; // wav file name
[x, fs, bits] = wavread(filename); // read wav file
fs, bits // display sampling frequency in Hz and quantization bit
dt = 1 / fs; // sampling interval in sec
t = [0: dt: (size(x,2) - 1) * dt]; // time in sec

subplot(2, 1, 1); // subdivision of graph area into 2x1 and present in area 1
plot2d(t, x(1,:)); // plot waveform
xlabel('Waveform', 'Time [s]', 'Waveform [a.u.]');

subplot(2, 1, 2); // subdivision of graph area into 2x1 and present in area 2

```

```
mapsound(x(1,:), 0.01, 200, 3000, 1, fs); //plot spectrogram
f=gcf();
f.color_map = hotcolormap(256); // change colormap
xlabel('Spectrogram', 'Time [s]', 'Frequency [Hz]');
```

### 3 実習内容

1. 正弦波およびチャープ信号に対して時間周波数解析を行う．各種パラメタ（特に周波数，時定数など）を変化させてスペクトログラムがどのように変化するか確認せよ．
2. 音声信号を使用して，スペクトログラムを描け．「おはよう」という発話のサンプルが5名分あるので，スペクトログラム上の際について観測せよ．

### 課題

演習の結果をレポートとしてまとめよ．なお，講義時間内に終了しなかった場合は IDE ラウンジもしくは自宅で完成させること．

締切は5月1日（月）10:30とし，講義開始前に回収する．前日までに提出する場合には，メールボックス S6-4（南6号館ロッカー室横）に提出のこと．表紙は必要ないが，大きさはA4判とし，学籍番号・氏名・提出日を上部欄外に記入すること．

### 参考文献

- [1] 声優のナレーション | フリー（無料）音源（音声）ダウンロード．[Online]． Available: <http://www.voice-pro.jp/announce/>
- [2] SCMPX. [Online]. Available: <http://www.din.or.jp/ch3/scmpx.html>